Program for Processing Standing Wave Measurements

PURPOSE This subroutine processes standing wave measurements with or without line loss on a transmission-line or waveguide load leading to its one-port characterization.

LANGUAGE FORTRAN IV.

AUTHOR J. W. Bandler, Department of Electrical Engineering, McMaster University, Hamilton, Ont., Canada.

AVAILABILITY Listing included in this description.

DESCRIPTION The subroutine to be described processes the actual experimental readings obtained from the well-known slotted-line standing wave measurement of a transmission-line or waveguide load and evaluates its one-port parameters. The load can be assumed to be either a complex or purely imaginary function of frequency and the measurements can be corrected for loss in the measuring system which would otherwise result in underestimating the standing wave ratio at the load. Variable data set reference numbers permit data to be read in or results to be printed out (optional) on any desired I/O unit.

The user calls the subroutine from his own program as follows.

CALL ONEPRT (F, Z, SWR, N, UL, FC,
TYPE, UNIT1, UNIT2, PRINT)

The variables in the argument list are defined as follows.

Output Variables

- F array of frequencies in GHz calculated from the measured data
- Z complex array of corresponding normalized load impedances calculated from the measured data
- SWR array of corresponding voltage standing wave ratios calculated from the measured data.

Input Variables

- N the number of sets of data to be read by the subroutine
- UL the location in cm of the load reference plane consistent with the scale on the slotted line
- FC the cutoff frequency in GHz of the waveguide (=0 for transmission lines)
- TYPE an integer which can be 1, 2, 3, 4, or 5 specifying the form of the data and defining the type of calculations to be made by the subroutine
- UNIT1 an integer specifying the data set reference number of the input unit
- UNIT2 an integer specifying the data set reference number of the output unit

Manuscript received January 20, 1969; revised April 1, 1969.

- PRINT a logical variable; when .TRUE. instructs the computer to print out results on the output unit; when .FALSE. instructs the computer not to print out any results (which are essentially contained in F, Z and SWR anyway).

The variables F, Z and SWR should be suitably dimensioned in the calling program (the number is up to the user) and also the variables TYPE, UNIT1, UNIT2 and PRINT should be appropriately defined.

Types of Measured Data

- TYPE = 1 specifies that the load is expected to be purely reactive (i.e., VSWR = ∞) so only positions about the standing wave minima are to be processed.
- TYPE = 2 specifies that the load is expected to be complex. In this case values of VSWR are to be processed as well as positions about the minima.
- TYPE = 3 is the same as TYPE = 2 but the VSWR readings are in dB.
- TYPE = 4 specifies that instead of measuring the VSWR directly the power ratio p in dB between the minimum and two corresponding points about the minimum was measured. The VSWR in this case is given by [1], [2]

$$\frac{\{\exp(0.23026p) - \cos^2(\pi d/\lambda_g)\}^{1/2}}{\sin \pi d/\lambda_g}$$

where d is the distance in cm between the corresponding points and λ_g is the wavelength in cm along the slotted line. This type is, therefore, particularly useful when the VSWR is large, say greater than 10, and when direct measurement may be difficult.

- TYPE = 5 is the same as TYPE = 4 except that line loss is to be taken into account for greater precision. Full details of the theory and measurement procedure for this type have been published [2].

Preparation of Measured Data

As shown in Fig. 1, integers are read by the subroutine in format I5. All other numbers are read in format F10.1 which, incidentally, allows the decimal point to be placed anywhere within a field of 10 characters.

The subroutine will read N sets of data. Each set must be preceded by a specification of the number of actual positions measured (= 2 × the number of standing wave minima); the number must lie between 4 and 24 inclusive. Following this line/card must be the actual readings in cm (8 per line/card) working sequentially down the scale towards the load without omitting any intermediate minima. If TYPE = 1 no further data is expected for this set. If TYPE = 2 or 3 then as many values of VSWR must follow on the next line/card as the number of minima employed (for averaging purposes). If TYPE = 4 only one value of p is expected on the next line/card. If TYPE = 5 proceed as for TYPE = 4 but an additional line/card is expected containing 1) the location of the attenuation reference plane, 2) the slotted-line attenuation in dB/cm, and 3) the

```
      SUBROUTINE ONEPRT(F, Z, SWR, N, UL, FC, TYPE, UNIT1, UNIT2,
     XPRINT)
      COMPLEX Z(1), Y
      INTEGER TYPE, RDNGS, SETS, UNIT1, UNIT2
      LOGICAL ODD, PRINT
      DIMENSION X(24), U(12), S(12), F(1), SWR(1)
      DATA C/29.97925/, PI, PIBY2/3.141593, 1.570796/, AVAL1,
     XAVAL2/.2302585, .1151293/
      IF (PRINT) WRITE (UNIT2, 1) N, UL, FC, TYPE
1     FORMAT ('1TRANSMISSION-LINE ONE-PORT PARAMETERS FROM EXPERIMENTAL
     XRESULTS'/'ON ='I4, 3X, 'UL ='G13.6, ' CM'3X, 'FC ='G16.9, ' GHZ'
     X3X, 'TYPE ='I2/'0    FREQ  LAMBDAG     VSWR       RHO   RETURN   TR
     XANSM        RES      REAC       CON      SUSC'/6X, 'GHZ        CM
     X                LOSS DB  LOSS DB      NLZD      NLZD      NLZD      NLZD'
     X)
      DO 14 L = 1, N
      F(L) = 0.
      Z(L) = 1.
      SWR(L) = 1.
      READ (UNIT1, 2) RDNGS
```

Fig. 1. FORTRAN IV listing of subroutine ONEPRT.

```
2     FORMAT (I5)
      SETS = RDNGS / 2
      IF (RDNGS .GE. 4 .AND. RDNGS .EQ. SETS + SETS .AND. RDNGS .LE. 24
     X) GO TO 4
      IF (PRINT) WRITE (UNIT2, 3)
3     FORMAT ('0THIS NUMBER OF READINGS CANNOT BE HANDLED')
      GO TO 14
4     ODD = SETS .NE. SETS / 2 * 2
      READ (UNIT1, 5) (X(I), I = 1, RDNGS)
5     FORMAT (8F10.2)
      UTHETA = 0.
      DO 6 I = 1, SETS
      J = I + I
      H = 0.5 * (X(J - 1) + X(J))
      U(I) = H
6     UTHETA = UTHETA + H
      AVLG = 2. * (U(1) - U(SETS)) / (SETS - 1)
      UBAR = UTHETA / SETS
      UTHETA = UBAR
      IF ( .NOT. ODD) UTHETA = UTHETA - 0.25 * AVLG
      DOVLG = (UTHETA - UL) / AVLG
      IF (TYPE .GT. 3) GO TO 8
      VSWR = 1.E4
      IF (TYPE .EQ. 1) GO TO 10
      READ (UNIT1, 5) (S(I), I = 1, SETS)
      VSWR = 0.
      DO 7 I = 1, SETS
7     VSWR = VSWR + S(I)
      VSWR = VSWR / SETS
      IF (TYPE .EQ. 3) VSWR = EXP(AVAL2 * VSWR)
      GO TO 10
8     READ (UNIT1, 5) P
      DBAR = 0.
      DO 9 I = 2, RDNGS, 2
9     DBAR = DBAR + X(I - 1) - X(I)
      DBAR = DBAR / SETS
      THETA = PI * DBAR / AVLG
      COSQ = COS(THETA)
      VSWR = SQRT(EXP(AVAL1 * P) - COSQ * COSQ) / SIN(THETA)
      IF (TYPE .NE. 5) GO TO 10
      READ (UNIT1, 5) UALFA, ASL, AT
      VSWR = VSWR * (1. + VSWR * ASL * AVAL2 * (UBAR - UALFA))
      VSWR = 1. / TANH(0.5 * ALOG((VSWR + 1.) / (VSWR - 1.)) - AVAL2 *
     XAT)
10    T = TAN(2. * PI * DOVLG)
      IF (VSWR .GE. 1.E4) GO TO 11
      Z(L) = CMPLX(1.,  - VSWR * T) / CMPLX(VSWR,  - T)
      RHO = (VSWR - 1.) / (VSWR + 1.)
      RL = - 20. * ALOG10(RHO)
      TL = - 10. * ALOG10(1. - RHO * RHO)
      GO TO 12
11    Z(L) = CMPLX(0.,  - T)
      RHO = 1.
      RL = 0.
      TL = 1.E4
      VSWR = 1.E4
12    SWR(L) = VSWR
      Y = 1. / Z(L)
      F(L) = SQRT(FC * FC + C * C / (AVLG * AVLG))
      IF (PRINT) WRITE (UNIT2, 13) F(L), AVLG, VSWR, RHO, RL, TL, Z(L),
     XY
13    FORMAT (10F9.4, 2X)
14    CONTINUE
      RETURN
      END
```

Fig. 1. (*Cont'd*).

```
      INTEGER TYPE
      COMPLEX Z(50)
      DIMENSION F(50), SWR(50)
1     READ (5, 2) N, UL, FC, TYPE
2     FORMAT (I5, 2F10.1, I5)
      CALL ONEPRT (F, Z, SWR, N, UL, FC, TYPE, 5, 6,  .TRUE. )
      GO TO 1
      END
```

```
    2     -0.1     6.55678       2
    6
17.17     16.97     15.19     14.99     13.20     13.00
 2.95      3.05      3.0
    6
15.97     15.37     14.14     13.52     12.30     11.69
 2.22      2.22      2.22
    1    -8.312       0.       5
    8
15.25     14.705    11.94     11.42      8.625     8.105     5.305     4.8
10.
12.        .0045      .1
```

```
TRANSMISSION-LINE ONE-PORT PARAMETERS FROM EXPERIMENTAL RESULTS

N =    2    UL =-0.100000E 00 CM    FC =  6.55677986     GHZ    TYPE = 2

    FREQ  LAMBDAG     VSWR        RHO   RETURN   TRANSM      RES      REAC       CON      SUSC
     GHZ       CM                      LOSS DB  LOSS DB     NLZD      NLZD      NLZD      NLZD
 10.0008   3.9700   3.0000    0.5000   6.0206   1.2494   1.1263    1.2139    0.4089   -0.4425
 10.4660   3.6750   2.2200    0.3789   8.4299   0.6730   1.7703    0.7704    0.4749   -0.2067
```

```
TRANSMISSION-LINE ONE-PORT PARAMETERS FROM EXPERIMENTAL RESULTS

N =    1    UL = -8.31200      CM    FC = 0.0               GHZ    TYPE = 5

    FREQ  LAMBDAG     VSWR        RHO   RETURN   TRANSM      RES      REAC       CON      SUSC
     GHZ       CM                      LOSS DB  LOSS DB     NLZD      NLZD      NLZD      NLZD
  4.5309   6.6167  14.0507    0.8671   1.2385   6.0536   0.0723   -0.1282    3.3400    5.9170
```

Fig. 2. Simple main program for calling subroutine ONEPRT, typical data and results for TYPE=2 and TYPE=5 [2].

total attenuation between the above reference plane and the load reference plane. See [2] for details on the meaning of these values and how to obtain them.

Evaluation and Presentation of Results

If PRINT=.TRUE. then the subroutine will print out results as indicated in Fig. 2. Whenever the VSWR is greater than or equal to 10^4, the load impedance is assumed to be purely imaginary, the reflection coefficient and return loss are taken as 1 and 0, respectively, and the standing wave ratio and transmission loss are set to 10^4 for convenience. Otherwise, the calculations are made in the conventional manner [2]. The output variables F, Z, and SWR may, of course, be used in further calculations which the user wishes to make.

The program has been tested on the IBM 360/65.

Acknowledgment

The author would like, in particular, to thank J. F. Wells of Mullard Research Laboratories, Redhill, Surrey, England, P. A. Macdonald and B. H. McDonald of the Electrical Engineering Department, University of Manitoba, Winnipeg, Canada, who contributed to the development of the program.

References

[1] A. B. Giordano, "Measurement of standing wave ratio," in *Handbook of Microwave Measurements*, 3rd ed., vol. 1, M. Sucher and J. Fox, Eds. New York: Wiley, 1963, ch. 2.

[2] J. W. Bandler, "Precision microwave measurement of the internal parasitics of tunnel-diodes," *IEEE Trans. Electron Devices*, vol. ED-15, pp. 275–282, May 1968.

Response Program for an Inhomogeneous Cascade of Rectangular Waveguides

PURPOSE This package calculates the input admittance versus frequency to an arbitrarily terminated inhomogeneous cascade of rectangular waveguides with or without junction discontinuity effects.

LANGUAGE FORTRAN IV.

AUTHORS J. W. Bandler, Department of Electrical Engineering, McMaster University, Hamilton, Ont., Canada, and P. A. Macdonald, Numerical Applications Group, Department of Electrical Engineering, University of Manitoba, Winnipeg, Man., Canada.

AVAILABILITY Listing included in this description.

DESCRIPTION This package of subprograms calculates the complex normalized input admittance versus frequency to an arbitrarily terminated homogeneous or inhomogeneous cascade of rectangular waveguides operating in the H_{10} mode. Discontinuity effects due to small *symmetrical* steps can be taken into account.

There is a LOGICAL FUNCTION subprogram (Fig. 1) which tests constraints: CUTOFF (M, A, B, FL, FU, PRINT, UNIT, BEWARE, RANGE, SMALLA, SMALLB).

Manuscript received February 21, 1969; revised April 28, 1969. This work was carried out at the University of Manitoba with financial assistance from the Faculty of Graduate Studies of the University of Manitoba and the National Research Council of Canada.